

In[217]:=

```
Clear["Global`*"]
```

In[218]:=

```

X = {-1, 2, 7/2, 4, 6}
F = {1, -1, 2, -2, 0}
body = Transpose[{X, F}]
n = Length[X] - 1
fobecny = Sum[a[i] * x^i, {i, 0, n}]
rovnice = Table[(fobecny /. {x -> X[[i]]}) == F[[i]], {i, 1, n + 1}]
reseni = Solve[rovnice][[1]]
f = fobecny /. reseni
Show[Plot[f, {x, -1, 6}], Graphics[{Red, Point[body]}]]

```

Out[218]=

$$\left\{-1, 2, \frac{7}{2}, 4, 6\right\}$$

Out[219]=

$$\{1, -1, 2, -2, 0\}$$

Out[220]=

$$\left\{\{-1, 1\}, \{2, -1\}, \left\{\frac{7}{2}, 2\right\}, \{4, -2\}, \{6, 0\}\right\}$$

Out[221]=

4

Out[222]=

$$a[0] + x a[1] + x^2 a[2] + x^3 a[3] + x^4 a[4]$$

Out[223]=

$$\left\{\begin{aligned} a[0] - a[1] + a[2] - a[3] + a[4] &= 1, & a[0] + 2 a[1] + 4 a[2] + 8 a[3] + 16 a[4] &= -1, \\ a[0] + \frac{7 a[1]}{2} + \frac{49 a[2]}{4} + \frac{343 a[3]}{8} + \frac{2401 a[4]}{16} &= 2, \\ a[0] + 4 a[1] + 16 a[2] + 64 a[3] + 256 a[4] &= -2, \\ a[0] + 6 a[1] + 36 a[2] + 216 a[3] + 1296 a[4] &= 0 \end{aligned}\right\}$$


Out[224]=

$$\left\{a[0] \rightarrow -\frac{979}{45}, a[1] \rightarrow -\frac{9337}{3780}, a[2] \rightarrow \frac{111367}{7560}, a[3] \rightarrow -\frac{38461}{7560}, a[4] \rightarrow \frac{353}{756}\right\}$$

Out[225]=

$$-\frac{979}{45} - \frac{9337 x}{3780} + \frac{111367 x^2}{7560} - \frac{38461 x^3}{7560} + \frac{353 x^4}{756}$$

Out[226]=



```
In[227]:= Together[InterpolatingPolynomial[body, x] - f]
```

```
Out[227]= 0
```

```
In[228]:= Lagrange[node_, i_, t_] := Simplify[InterpolatingPolynomial[
  Table[{node[[j]], KroneckerDelta[i, j]}, {j, 1, Length[node]}], t]]
```

```
In[229]:= l = Table[Lagrange[X, i, x], {i, 1, n + 1}]
Plot[l, {x, -1, 6}]
```

```
Out[229]= {
   $\frac{1}{945} (336 - 404 x + 172 x^2 - 31 x^3 + 2 x^4),$ 
   $-\frac{1}{72} (1 + x) (-168 + 118 x - 27 x^2 + 2 x^3),$ 
   $\frac{16}{135} (-2 + x) (1 + x) (24 - 10 x + x^2),$ 
   $-\frac{1}{20} (-6 + x) (-2 + x) (1 + x) (-7 + 2 x),$ 
   $\frac{1}{280} (-4 + x) (-2 + x) (1 + x) (-7 + 2 x)$ 
}
```

```
Out[230]=
```



```
In[231]:= Together[f - Sum[F[[i]] * l[[i]], {i, 1, n + 1}]]
```

```
Out[231]= 0
```

```
In[232]:= (* věta o chybě lineární Lagrangeovy interpolace *)
```

In[233]:=

```
g = x ^ 2  
a = 0  
f = b * x  
r = g - f  
D[r, x]  
Chyba = -r /. {x -> b / 2}  
Chyba /. {b -> 0.01}
```

Out[233]=

 x^2

Out[234]=

0

Out[235]=

 $b x$

Out[236]=

 $-b x + x^2$

Out[237]=

 $-b + 2 x$

Out[238]=

 $\frac{b^2}{4}$

Out[239]=

0.000025

In[240]:=

```

g = Sqrt[x]
a = 0
f = (1 / Sqrt[b]) * x
r = g - f
Factor[D[r, x]]
Chyba = r /. {x -> b / 4}
Chyba /. {b -> 0.01}

```

Out[240]=

$$\sqrt{x}$$

Out[241]=

0

Out[242]=

$$\frac{x}{\sqrt{b}}$$

Out[243]=

$$\sqrt{x} - \frac{x}{\sqrt{b}}$$

Out[244]=

$$\frac{\sqrt{b} - 2\sqrt{x}}{2\sqrt{b}\sqrt{x}}$$

Out[245]=

$$\frac{\sqrt{b}}{4}$$

Out[246]=

0.025

In[247]:=

```

Clear["Global`*"]
f = Sum[a[i] * x^i, {i, 0, 2}]

```

Out[248]=

$$a[0] + x a[1] + x^2 a[2]$$

In[249]:=

```

f /. Solve[
  {(f /. {x -> 1}) == 9, (D[f, x] /. {x -> 2}) == 11, (D[f, {x, 2}] /. {x -> 4}) == 4}][[1]]

```

Out[249]=

$$4 + 3x + 2x^2$$